
FORCST(COEF=ベクトル名, DEPVAR=変数名, DYNAM, PRINT,
RHO=スカラー, STATIC) 予測変数名 [独立変数のリスト];

機能：

FORCST は、TSP プログラムの中で実行される線形方程式推定の結果を使って、従属変数の予測値を計算するコマンドです。標本期間は、推定に使われた期間と同じである必要はありません。予測計算には、別の外生変数のセットを使うこともできます。予測に用いるモデルは、以前の線形推定プロシジャー (OLSQ, INST, AR1) で指定したものでよいし、あるいは様々なオプションを用いて任意に定めることも可能です。LSQ や FIML などの非線形推定プロシジャーを行なった後に予測値を計算するには、GENR, SIML, SOLVE コマンドを使います。もしモデルが BJEST によって、ボックス・ジェンキンスの方法でモデルの推定を行なっているのなら、BJFRCSST によって予測計算を行なって下さい。

使用法:

FORCST の最も簡単な使用法は、予測に使うモデルを定めた推定コマンドにつづけます。途中で他の推定法が入ることはゆるされません。もし推定と予測で異なった標本期間を使いたいのなら、推定と予測の間で SMPL ステートメントで標本期間を変更して下さい。この最も簡単な場合でのコマンドの使い方は

FORCST 予測した変数に与える名前 ;

となります。結果をプリントとプロットしたい場合には、PRINT オプションを指定して下さい。

FORCST のもう一つの使用法では、推定に続いてすぐ FORCST を指定する必要はありません。しかし係数ベクトル、右辺の変数の名前、そして系列相関を修正したい場合は RHO の値を指定する必要があります。

もし STATIC や NODYNAM が指定されていれば、FORCST は予測値の計算ではラグ付き従属変数を他の外生変数と同様に扱います。しかし FORCST ステートメントでもし (デフォルトの) DYNAM オプションを指定すると、ラグ付き従属変数には計算された従属変数の値がダイナミックに代入されます。初期状態では、実際のラグ付従属変数が第一期に使われます。

AR1 の回帰に続いて FORCST を使用するばあいは、回帰式の計算結果に含まれる RHO の計算値を使って、系列相関に対する修正を行ないながら予測値を計算します。このためには、

$$\hat{y}_t = X_t b \quad (t = 1 \text{ の時}), \quad \hat{y}_t = X_t b - \rho X_{t-1} b + \rho y_{t-1} \quad (t > 1 \text{ の時})$$

が計算できるように、従属変数の値がプログラムで利用できなければいけません。もし STATIC が指定されているならばこの「静学的な」予測値が計算されます。もし本当の動学的な予測、あるいは外挿が必要ならば、標本期間を Y のデータの最後の観測値から初めて、(デフォルトである) DYNAM オプションを使ってください。動学の場合でも静学の場合でも、AR1 に続いて FORCST がすぐ指定されない場合は、従属変数の名前が DEPVAR= オプションで指定されていないといけません。

もし SMPL ベクトルに飛び超しがあるならば、FORCST は飛び越し前と後の標本を別個に扱います。もし DYNAM オプションが指定されていないければ、飛び越しがある場合でもない場合でも予測値は同じです。もし RHO が 0 でないか DYNAM オプションが指定されている場合は、予測計算は飛び越しのあとで新た

に始まります。したがって、飛び越しのあとで新しい初期値が必要です。結果として、飛び越しがある場合と無い場合では、予測結果に違いが生じます。

オプション:

COEF= 予測計算で使われる係数推定値を含むベクトルの名前を指定します。このベクトルを、すでに行なわれた推定結果において得られた@COEFという名前で置き換えることも可能です。このベクトルにおける係数の順序は、予測モデルに含まれる説明変数の名前の順序と一致しないといけません。

DEPVAR= 推定モデルでの従属変数の名前です。もし動学的な予測において、推定プロシジャーに続いてFORCSTプロシジャーがすぐ行なわれない場合は、このオプションの指定がなければ正しいダイナミック予測が行えません。

DYNAM/STATIC 予測が動学的なものか静学的なものを定めます。静学的な場合では、ラグ付き内生変数の(与えられた)過去の値を予測期間全期について使います。一方動学的な予測では、計算された予測値をラグ付内生変数の値として使います。(すなわち、2番目の観測値をラグ1に、3番目の観測値をラグ2の変数に、等々とします) AR(1) 予測において残差に現われるラグ付き内生変数に関しても、このような処理はほどこされます。もちろんラグ付き従属変数が使われず、AR1 予測が指定されていない場合には、STATICが標準となります。

PRINT/NOPRINT 予測結果をプリントとプロットすべきかどうかを指定します。もしこのオプションが指定されていれば、タイトルと、予測モデルについてプリントされます。

RHO= AR1 を使った予測が必要で、直前の推定結果を利用しない場合は、このオプションによって系列相関パラメータの値を指定します。

例:

```
SMPL 1 20;
OLSQ CONS C GNP;
SMPL 21 30;
FORCST (PRINT) CONSP;
```

この例では、OLSQ 方程式は将来期間に外挿をするために使われます。すなわち外生変数の値は、この外挿期間にわたって与えられていないといけません。CONS の外挿値は CONSP という名前で保存されます。(PRINT オプションが選択されているので) 予測値もプリントされかつプロットされます。

```
SMPL 1 20;
OLSQ CONS C GNP;
COPY @COEF B;
.
.
FORCST (COEF=B) CONSP C GNPNEW;
```

この例では OLSQ で推定された式を使って CONSP を計算しますが、回帰式の右辺GNP とは別の系列となります。もしこのタイプの AR1 予測が欲しいときには、系列相関係数を括弧内で RHO= オプションとして指定します。

```
SMPL 1 20;
OLSQ I I(-1) GNP;
SMPL 21 30;
FORCST (PRINT, DYNAM) IFIT;
```

FORCST

この例では, DYNAM オプションを選択して, 回帰式の右辺に含まれるラグ付内生変数 $I(-1)$ に予測値を代入し, 動学的な予測をします.

```
SMPL 1 20;
AR1 I I(-1) GNP;
SET R=@RHO ; MFORM B=@COEF ;
SMPL 20 30;
FORCST (PRINT,DYNAM) IFIT;
.
.
(他のTSPプログラムがきます.)
.
.
SMPL 1 20;
FORCST (COEF=B,RHO=R) I2FIT I(-1) GNPNEW;
```

この例ではFORCSTプロシジャーの動学的な使い方と, 静学的な使い方の両方を示しています. 最初のプログラムでは動学的な指定がされています. R と B を保存するプログラム文は後半でのみ必要です. 後半では推定と同じ期間にわたって予測をしようとしています, 別の右辺の変数 GNPNEW が使われています.

```
SMPL 65:1 82:4 ;
AR1 SALES C SALES(-1) GNP GNP(-1);
SMPL 82:4 86:4;
GENR GNP1=GNPFCST;
GENR GNP2=1.1*GNPFCST;
GENR GNP3=0.9*GNPFCST;
DOT 1 2 3;
FORCST (DYNAM,PRINT) SALESF. C SALES(-1) GNP. GNP.(-1);
ENDDOT;
```

この例では, 3つの異なった GNP 予測を基に, 販売量変数の予測値を計算する方法を示しています.

アウトプット:

プリント・オプションが指定されていなければ, FORCST は計算結果をプリントしません. 結果は予測値の系列だけがデータ領域内に保存されます.

PRINT が指定されていると, タイトル, 予測に使われる係数ベクトル, 系列相関係数, そして予測が動学的か, 静学的かをプリントします. 予測値のプロットもプリントされます. グラフでは, 左側に変数名がプリントされ, 右側に系列変数の値がプリントされます. 予測値の系列も保存されます.

参考文献:

Pindyck, Robert S. and Daniel L. Rubinfeld, **Econometric Models and Economic Forecasts**, Chapter 6, McGraw-Hill Book Company, New York, 1976.